

The TPTP Extended Typed First-order Form - TFX

Geoff Sutcliffe¹ and Evgenii Kotelnikov²

¹ University of Miami, USA

² Chalmers University of Technology, Sweden

Abstract

1 Introduction

The TPTP world [11] is a well established infrastructure that supports research, development, and deployment of Automated Theorem Proving (ATP) systems for classical logics. The TPTP world includes the TPTP problem library, the TSTP solution library, the TMTP model library, standards for writing ATP problems and reporting ATP solutions, tools and services for processing ATP problems and solutions, and it supports the CADE ATP System Competition (CASC). Various parts of the TPTP world have been deployed in a range of applications, in both academia and industry. The web page <http://www.tptp.org> provides access to all components.

The TPTP language is one of the keys to the success of the TPTP world. The language is used for writing both TPTP problems and TSTP solutions, which enables convenient communication between different systems and researchers. Originally the TPTP world supported only first-order clause normal form (CNF) [14]. Over the years support for full first-order form (FOF) [10], monomorphic typed first-order form (TF0) [13], rank-1 polymorphic typed first-order form (TF1) [2], monomorphic typed higher-order form (TH0) [12], and rank-1 polymorphic typed higher-order form (TH1) [4], have been added. (TF0 and TF1 together form the TFF language family; TH0 and TH1 together form the THF language family.)

Since the inception of TFF and THF there have been some features that have received little use, and hence little attention. In particular, tuples, conditional expressions (if-then-else), and let expressions (let-defn-in) were neglected, and in TFF they were horribly formulated with variants to distinguish between their use with formulae and terms. Recently, conditional expressions and let expressions have become more important because of their use in software verification applications. In an independent development, Evgenii Kotelnikov et al. introduced the FOOL logic [7], which extends first-order logic so that (i) formulae can be used as terms of the boolean type, (ii) variables of the boolean type can be used as formulae, (iii) tuple terms and tuple types are available as first-class citizens, and (iv) conditional and let expressions are supported. This logic can be automatically translated to ordinary many-sorted first-order logic [7]. Features of FOOL can be used to concisely express problems coming from program analysis [8] or translated from more expressive logics. Conditional expressions and let expressions of FOOL resemble those of the SMT-LIB language version 2 [1].

The TPTP's new Typed First-order form eXtended (TFX) language remedies the old weaknesses of TFF, and incorporates the features of FOOL logic. This has been achieved by conflating (with some exceptions) formulae and terms, simplifying tuples in plain TFF, including fully expressive tuples in TFX, removing the old conditional expressions and let expressions from TFF, and including new elegant forms of conditional expressions and let expressions as part of TFX. These more elegant forms have been mirrored in THF. TFX is a superset of the TFF language.

This paper describes the extensions to the TFF language form that define the TFX language, and changes to the THF language that correspond to decisions made for TFX. The remainder of

this paper is organized as follows: Section 2 reviews the TFF and THF languages, and describes the FOOL logic. Section 3 provides technical and syntax details of the new features of TFX. Section 4 describes the evolving software support for TFX, and provides some examples that illustrate its use. Section 5 concludes.

2 The TPTP Language and the FOOL Logic

The TPTP language is a human-readable, easily machine-parsable, flexible and extensible language, suitable for writing both ATP problems and solutions. The top level building blocks of the TPTP language are *annotated formulae*. An annotated formula has the form:

language(name, role, formula, [source, [useful_info]]).

The *languages* supported are clause normal form (**cnf**), first-order form (**fof**), typed first-order form (**tff**), and typed higher-order form (**thf**). The *role*, e.g., **axiom**, **lemma**, **conjecture**, defines the use of the formula in an ATP system. In the *formula*, terms and atoms follow Prolog conventions, i.e., functions and predicates start with a lowercase letter or are 'single quoted', variables start with an uppercase letter, and all contain only alphanumeric characters and underscore. The TPTP language also supports interpreted symbols, which either start with a \$, or are composed of non-alphanumeric characters, e.g., the truth constants **\$true** and **\$false**, and integer/rational/real numbers such as 27, 43/92, -99.66. The basic logical connectives are **!**, **?**, **~**, **|**, **&**, **=>**, **<=**, **<=>**, and **<~>**, for \forall , \exists , \neg , \vee , \wedge , \Rightarrow , \Leftarrow , \Leftrightarrow , and \oplus respectively. Equality and inequality are expressed as the infix operators **=** and **!=**. An example annotated first-order formula, supplied from a file, is:

```
fof(union,axiom,
  ( ! [X,A,B] :
    ( member(X,union(A,B))
      <=> ( member(X,A)
          | member(X,B) ) )
    file('SET006+0.ax',union),
    [description('Definition of union'), relevance(0.9)]).
```

2.1 The Typed First-order Form TFF

TFF extends the basic FOF language with *types* and *type declarations*. The TF0 variant is monomorphic, and the TF1 variant is rank-1 polymorphic. Every function and predicate symbol is declared before its use, with a *type signature* that specifies the types of the symbol's arguments and result. TF0 types τ have the following forms:

- the predefined types **\$i** for ι (individuals) and **\$o** for o (booleans);
- the predefined arithmetic types **\$int** (integers), **\$rat** (rationals), and **\$real** (reals);
- user-defined types (constants).

User-defined types are declared (before their use) to be of the kind **\$tType**, in annotated formulae with a **type** role – see Figure 1 for examples. TF0 type signatures ς have the following forms:

- individual types τ ;
- $(\tau_1 * \dots * \tau_n) > \tilde{\tau}$ for $n > 0$, where τ_i are the argument types, and $\tilde{\tau}$ is the result type.

The type signatures of uninterpreted symbols are declared like types, in annotated formulae with a **type** role – see Figure 1 for examples. The type of `=` is ad hoc polymorphic over all types except `$o` (this restriction is lifted in TFX), with both arguments having the same type and the result type being `$o`. The types of arithmetic predicates and functions are ad hoc polymorphic over the arithmetic types; see [13] for details. Figure 1 illustrates some TF0 formulae, whose conjecture can be proved from the axioms (it is the TPTP problem PUZ130_1.p).

```
%-----
tff(animal_type,type, animal: $tType ).
tff(cat_type,type, cat: $tType ).
tff(dog_type,type, dog: $tType ).
tff(human_type,type, human: $tType ).
tff(cat_to_animal_type,type, cat_to_animal: cat > animal ).
tff(dog_to_animal_type,type, dog_to_animal: dog > animal ).
tff(garfield_type,type, garfield: cat ).
tff(odie_type,type, odie: dog ).
tff(jon_type,type, jon: human ).
tff(owner_of_type,type, owner_of: animal > human ).
tff(chased_type,type, chased: ( dog * cat ) > $o ).
tff(hates_type,type, hates: ( human * human ) > $o ).

tff(human_owner,axiom, ! [A: animal] : ? [H: human] : H = owner_of(A) ).
tff(jon_owns_garfield,axiom, jon = owner_of(cat_to_animal(garfield)) ).
tff(jon_owns_odie,axiom, jon = owner_of(dog_to_animal(odie)) ).
tff(jon_owns_only,axiom,
  ! [A: animal] :
    ( jon = owner_of(A)
      => ( A = cat_to_animal(garfield) | A = dog_to_animal(odie) ) ) ).

tff(dog_chase_cat,axiom,
  ! [C: cat,D: dog] :
    ( chased(D,C)
      => hates(owner_of(cat_to_animal(C)),owner_of(dog_to_animal(D))) ) ).
tff(odie_chased_garfield,axiom, chased(odie,garfield) ).

tff(jon_hates_jon,conjecture, hates(jon,jon) ).
%-----
```

Figure 1: TF0 Formulae

The polymorphic TF1 extends TF0 with (user-defined) *type constructors*, *type variables*, polymorphic symbols, and one new binder. TF1 types τ have the following forms:

- the predefined types `$i` and `$o`;
- the predefined arithmetic types `$int`, `$rat`, and `$real`;
- user-defined n -ary type constructors applied to n type arguments;
- type variables, which must be quantified by `!>` – see the type signature forms below.

Type constructors are declared (before their use) to be of the kind $(\$tType * \dots * \$tType) > \$tType$, in annotated formulae with a **type** role. TF1 type signatures ς have the following forms:

- individual types τ ;
- $(\tau_1 * \dots * \tau_n) > \tilde{\tau}$ for $n > 0$, where τ_i are the argument types and $\tilde{\tau}$ is the result type (with the same caveats as for TF0);
- $!>[\alpha_1 : \text{\$tType}, \dots, \alpha_n : \text{\$tType}] : \varsigma$ for $n > 0$, where $\alpha_1, \dots, \alpha_n$ are distinct type variables and ς is a type signature.

The binder $!>$ in the last form denotes universal quantification in the style of $\lambda\Pi$ calculi. It is only used at the top level in polymorphic type signatures. All type variables must be of type $\text{\$tType}$; more complex type variables, e.g., $\text{\$tType} > \text{\$tType}$ are beyond rank-1 polymorphism. As in TF0, arithmetic symbols and equality are ad hoc polymorphic. An example of TF1 formulae can be found in [4].

2.2 The FOOL Logic

FOOL [7], standing for First-Order Logic (FOL) + bOoleans, is a variation of many-sorted first-order logic in which (i) formulae can be used as terms of the boolean type, (ii) variables of the boolean type can be used as formulae, (iii) tuple terms and tuple types are available as first-class citizens, and (iv) conditional and let expressions are supported. FOOL can be straightforwardly extended with the polymorphic theory of tuples [8]. This theory includes first class tuples type and first class tuple terms. In the sequel we consider such extension. There is a model-preserving transformation of FOOL formulae to FOL formulae [7], so that an implementation of the transformation makes it possible to prove FOOL formulae using a regular first-order theorem prover. Formulae of FOOL can also be efficiently translated to a first-order clausal normal form [6]. The following describes these features of FOOL, illustrating them using examples taken from [5] and [8].

Boolean terms

FOOL contains a fixed two-element type *bool*, allows quantification over variables of type *bool*, and considers terms of type *bool* to be formulae. Formula 1 is a syntactically correct tautology in FOOL.

$$(\forall x : \text{bool})(x \vee \neg x) \quad (1)$$

The existence of *bool* terms means that a function or a predicate can take a formula as an argument, and formulae can be used as arguments to equality. For example, logical implication can be defined as a binary function *impl* of the type $\text{bool} \times \text{bool} \rightarrow \text{bool}$ using the following axiom.

$$(\forall x : \text{bool})(\forall y : \text{bool})(\text{impl}(x, y) \Leftrightarrow \neg x \vee y). \quad (2)$$

Formula 2 can be equivalently expressed with \doteq instead of \Leftrightarrow . Then it is possible to express that *P* is a graph of a (partial) function of the type $\sigma \rightarrow \tau$ as follows.

$$(\forall x : \sigma)(\forall y : \tau)(\forall z : \tau)(\text{impl}(P(x, y) \wedge P(x, z), y \doteq z)) \quad (3)$$

Tuples

For all types $\sigma_1, \dots, \sigma_n$, $n > 0$ FOOL contains a type $(\sigma_1, \dots, \sigma_n)$ of the *n*-ary tuple. Each type $(\sigma_1, \dots, \sigma_n)$ is considered first class, that is, it can be used as argument of a function or predicate symbol and in a quantifier. An expression (t_1, \dots, t_n) , where t_1, \dots, t_n are terms of types $\sigma_1, \dots, \sigma_n$, respectively, is a term of the type $(\sigma_1, \dots, \sigma_n)$.

Tuples are ubiquitous in mathematics and programming languages. For example, one can use the tuple sort (\mathbb{R}, \mathbb{R}) as the sort of complex numbers. Thus, the term $(2, 3)$ represents the complex number $2 + 3i$. A function symbol *plus* which represents addition of complex numbers has the type $(\mathbb{R}, \mathbb{R}) \times (\mathbb{R}, \mathbb{R}) \rightarrow (\mathbb{R}, \mathbb{R})$.

Conditional expressions

FOOL contains expressions of the form **if** ψ **then** s **else** t , where ψ is a formula, and s and t are terms of the same type. The semantics of such expressions mirrors the semantics of conditional expressions in programming languages, and are therefore convenient for expressing formulae coming from program analysis. For example, consider the *max* function of the type $\mathbb{Z} \times \mathbb{Z} \rightarrow \mathbb{Z}$ that returns the maximum of its arguments. Its definition can be expressed in FOOL as

$$(\forall x : \mathbb{Z})(\forall y : \mathbb{Z})(\text{max}(x, y) \doteq \text{if } x \geq y \text{ then } x \text{ else } y). \quad (4)$$

FOOL allows if-then-else expressions to occur as formulae, as in the following valid property of *max*

$$(\forall x : \mathbb{Z})(\forall y : \mathbb{Z})(\text{if } \text{max}(x, y) \doteq x \text{ then } x \geq y \text{ else } y \geq x). \quad (5)$$

Let expressions

FOOL contains expressions of the form **let** $D_1; \dots; D_k$ **in** t , where $k > 0$, t is either a term or a formula, and D_1, \dots, D_k are simultaneous non-recursive definitions. FOOL allows definitions of function symbols, predicate symbols, and tuples.

The definition of a function symbol $f : \sigma_1 \times \dots \times \sigma_n \rightarrow \tau$ has the form $f(x_1 : \sigma_1, \dots, x_n : \sigma_n) = s$, where $n \geq 0$, and s is a term of the type τ . The following let expression denotes the maximum of three integer constants a , b , and c using a local definition of the function symbol *max*.

$$\begin{aligned} \text{let } \text{max}(x : \mathbb{Z}, y : \mathbb{Z}) &= \text{if } x \geq y \text{ then } x \text{ else } y \\ \text{in } \text{max}(\text{max}(a, b), c) \end{aligned} \quad (6)$$

The definition of a predicate symbol $p : \sigma_1 \times \dots \times \sigma_n$ has the form $p(x_1 : \sigma_1, \dots, x_n : \sigma_n) = \phi$, where $n \geq 0$, and ϕ is a formula. The following let expression denotes equivalence of two boolean constants A and B using a local definition of the predicate symbol *impl*.

$$\begin{aligned} \text{let } \text{impl}(x : \text{bool}, y : \text{bool}) &= \neg x \vee y \\ \text{in } \text{impl}(A, B) \wedge \text{impl}(B, A) \end{aligned} \quad (7)$$

The definition of a tuple has the form $(c_1, \dots, c_n) = s$, where $n > 1$, c_1, \dots, c_n are constant symbols of the types $\sigma_1, \dots, \sigma_n$, respectively, and s is a term of the type $(\sigma_1, \dots, \sigma_n)$. The following formula defines addition for complex numbers using a local definition of a tuple.

$$\begin{aligned} (\forall x : (\mathbb{R}, \mathbb{R}))(\forall y : (\mathbb{R}, \mathbb{R})) \\ (\text{plus}(x, y) \doteq \text{let } (a, b) = x; (c, d) = y \text{ in } (a + c, b + d)). \end{aligned} \quad (8)$$

The main application of let expressions with tuple definitions is in problems coming from program analysis, namely modelling of assignments [8]. The lefthand side of Figure 2 shows an example of an imperative **if** statement containing assignments to integer variables, and an **assert** statement. This can be encoded as a FOOL formula as shown on the righthand side, using let expressions with definitions of tuples that capture the assignments.

<pre> if (x > y) { t := x; x := y; y := t; } assert x <= y; </pre>	<pre> let (x,y,t) = if x > y then let t = x in let x = y in let y = t in (x,y,t) else (x,y,t) in x ≤ y </pre>
--	--

Figure 2: FOOL encoding of an if statement

The semantics of let expressions in FOOL mirrors the semantics of simultaneous non-recursive local definitions in programming languages. That is, neither of the definitions D_1, \dots, D_n uses function or predicate symbols created by any other definition. In the following example, constants a and b are swapped by a let expression. The resulting formula is equivalent to $f(b, a)$.

$$\text{let } a = b; b = a \text{ in } f(a, b) \quad (9)$$

3 The TFX Syntax

The TPTP syntax has been extended to provide the features of FOOL logic, and at the same time some of the previous weaknesses have been remedied. This has been achieved by conflating (with some exceptions) formulae and terms. Tuples have been simplified in TFF, and fully expressive tuples included in TFX. The old conditional expressions and let expressions from TFF have been removed, and have been included elegantly as part of TFX.

Boolean terms

Formulae are permitted as terms. For example ...

```

tff(p_type,type,p: ($i * $o * $int) > $o ).
tff(q_type,type,q: ($int * $i) > $o ).
tff(me_type,type,me: $i ).
tff(fool_1,axiom,! [X: $int] : p(me,! [Y: $i] : q(X,Y),27) ).

```

... and ...

```

tff(p_type,type,p: $i > $o ).
tff(q_type,type,q: $o > $o ).
tff(me_type,type,me: $i ).
tff(fool_2,axiom,q((~p(me)) != q(me)) ).

```

A consequence of allowing formulae as terms is that the default typing of functions and predicates supported in plain TFF (functions default to $(\$i * \dots * \$i) > \$i$ and predicates default to $(\$i * \dots * \$i) > \$o$) is not supported in TFX (as is the case for THF).

Note that not all terms can be used as formulae. Tuples, numbers, and “distinct objects” cannot be used as formulae.

Boolean variables as formulae

Variables of type `$o` can be used as formulae. For example ...

```
tff(implies_type,type,implies: ($o * $o) > $o).
tff(implies_defn,axiom, ! [X: $o,Y: $o]: (implies(X,Y) <=> ((~X) | Y)) ).
```

Tuples

Tuples in TFF are written in `[]` brackets, and can contain non-boolean terms; tuples of formulae have been removed from TFF. Tuples in TFX are written in `[]` brackets, and can contain any type of term (including formulae and variables of type `$o`). Signatures can contain tuple types. For example ...

```
tff(t1_type,type,t1: $tType ).
tff(d_type,type,d: [$i,t1,$int] ).
tff(p_type,type,p: [t1,$i] > $o ).
tff(f_type,type,f: ($o * [$i,t1,$int]) > [t1,$i] ).
tff(tuples_2,axiom,p(f($true,d)) ).
```

Tuples can occur only as terms (not as formulae, and not on the lefthand side of a type declaration). Tuples can occur anywhere they are well-typed. For example ...

```
tff(p_type,type,p: ([$int,$i,$o] * $o * $int) > $o ).
tff(q_type,type,q: ($int * $i) > $o ).
tff(me_type,type,me: $i ).
tff(tuples_1,axiom,! [X: $int] : p([33,me,$true],! [Y: $i] : q(X,Y),27) ).
```

Tuples cannot be typed. Rather the elements must be typed separately. For example ...

```
tff(a_type,type,a: $int).
tff(b_type,type,b: $int).
tff(p_type,type,p: [$int,$int] > $o).
tff(p,axiom,p([a,b])).
```

... cannot be abbreviated to ...

```
tff(ab_type,type,[a,b]: [$int,$int]).
tff(p_type,type,p: [$int,$int] > $o).
tff(p,axiom,p([a,b])).
```

Conditional expressions

Conditional expressions are parametric polymorphic, taking a formula as the first argument, then two terms or formulae of any one type as the second and third arguments, as the true and false return values respectively, i.e., the return type is the same as that of the second and third arguments. For example ...

```
tff(p_type,type,p: $int > $o).
tff(q_type,type,q: $int > $o).
tff(max_type,type,max: ($int * $int) > $int).
tff(ite_1,axiom,! [X:$int,Y:$int] : $ite($greater(X,Y),p(X),p(Y)) ).
tff(ite_2,axiom,! [X:$int,Y:$int] : q($ite($greater(X,Y),X,Y)) ).
tff(max_defn,axiom,
```

```

    ! [X: $int,Y: $int]: (max(X,Y) = $ite($greatereq(X,Y),X,Y)) ).
tff(max_property,axiom,
    ! [X: $int,Y: $int]:
        $ite(max(X,Y) = X,$greatereq(X,Y),$greatereq(Y,X)) ).

```

Tuples can be used usefully in conditional expressions. For example ...

```

tff(p_type,type,p: [$int,$int] > $o).
tff(d_type,type,d: [$int,$int]).
tff(ite_3,axiom,
    ! [X:$int,Y:$int] : p($ite($greater(X,Y),[X,Y],[Y,X])) ).
tff(ite_4,axiom,
    ! [X:$int,Y:$int] : (d = $ite($greater(X,Y),[X,Y],[Y,X])) ).

```

Let expressions

Let expressions provide the types of defined symbols, definitions for the symbols, and a formula/term in which the definitions are applied. Each type declaration is the same as a type declaration in an annotated formula with the type role, and multiple type declarations are given in []ed tuples of declarations. Each definition defines the expansion of one of the declared symbols, and multiple definitions are given in []ed tuples of definitions. For example ...

```

tff(p_type,type,p: $int > $o ).
tff(let_1,axiom,$let(c: $int,c:= 27,p(c)) ).

```

... is equivalent to ...

```

tff(p_type,type,p: $int > $o ).
tff(let_1,axiom,p(27) ).

```

If variables are used in the lefthand side of a definition, their values are supplied in the defined symbol's use. Such variables do not need to be declared (they are implicitly declared to be of the type defined by the symbol declaration), but must be top-level arguments of the defined symbol and be pairwise distinct. For example ...

```

tff(i_type,type,i: $int).
tff(j_type,type,j: $int).
tff(f_type,type,f: ($int * $int * $int * $int) > $rat).
tff(p_type,type,p: $rat > $o ).
tff(let_2,axiom,
    $let(ff: ($int * $int) > $rat, ff(X,Y):= f(X,X,Y,Y), p(ff(i,j))) ).

```

... is equivalent to ...

```

tff(i_type,type,i: $int).
tff(j_type,type,j: $int).
tff(f_type,type,f: ($int * $int * $int * $int) > $rat).
tff(p_type,type,p: $rat > $o ).
tff(let_2,axiom,p(f(i,i,j,j)) ).

```

The defined symbols have scope over the formula/term in which the definitions are applied, shadowing any definition outside the let expression. The RHS of a definition can have symbols with the same name as the defined symbol, but refer to symbols defined outside the let expression. For example ...


```

tff(array_type,type,array: $int > $real).
tff(p_type,type,p: $real > $o).
tff(let_3,axiom,
    $let(array: $int > $real,
        array(I):= $ite(I = 3,5.2,array(I)),
        p($sum(array(2),array(3))) ) ).

```

... is equivalent to ...

```

tff(array_type,type,array: $int > $real).
tff(p_type,type,p: $real > $o).
tff(let_3,axiom,p($sum(array(2),5.2)) ).

```

The occurrence of `array` in the lefthand side of the definition `array(I):=`, and the occurrences in the formula in which the `let` expression is applied `p($sum(array(2),array(3)))`, are the defined symbol. The occurrence in the righthand side of the definition `$ite(I = 3,5,array(I))` is the globally defined symbol. Tuples can be used to define multiple symbols in `let` expressions in parallel (not sequential, i.e., this is `let`, not `let*`). For example ...

```

tff(a_type,type,a: $int).
tff(b_type,type,b: $int).
tff(p_type,type,p: ($int * $int) > $o).
tff(let_tuple_1,axiom,$let([a: $int,b: $int],[a:= b,b:= a],p(a,b)) ).

```

... is equivalent to ...

```

tff(a_type,type,a: $int).
tff(b_type,type,b: $int).
tff(p_type,type,p: ($int * $int) > $o).
tff(let_tuple_1,axiom,p(b,a)).

```

... and ...

```

tff(i_type,type,i: $int).
tff(f_type,type,f: ($int * $int * $int * $int) > $int).
tff(p_type,type,p: $int > $o ).
tff(let_tuple_2,axiom,
    $let([ff: ($int * $int) > $int, gg: $int > $int],
        [ff(X,Y):= f(X,X,Y,Y), gg(Z):= f(Z,Z,Z,Z)],
        p(ff(i,gg(i)))) ).

```

... is equivalent to ...

```

tff(i_type,type,i: $int).
tff(f_type,type,f: ($int * $int * $int * $int) > $int).
tff(p_type,type,p: $int > $o ).
tff(let_tuple_2,axiom,p(f(i,i,f(i,i,i,i),f(i,i,i,i))) ).

```

As tuples cannot be typed ...

```

tff(p_type,type,p: ($int * $int) > $o).
tff(not_let_tuple,axiom,
    $let([a: $int,b: $int],[a:= b,b:= a],p(a,b)) ).

```

... cannot be abbreviated to ...

```
tff(p_type,type,p: ($int * $int) > $o).
tff(not_let_tuple,axiom,
    $let([a,b]: [$int,$int],[a:= b,b:= a],p(a,b)) ).
```

Sequential let expressions (let*) can be implemented by nesting. For example ...

```
tff(i_type,type,i: $int).
tff(f_type,type,f: ($int * $int * $int * $int) > $int).
tff(p_type,type,p: $int > $o ).
tff(let_tuple_3,axiom,
    $let(ff: ($int * $int) > $int,
        ff(X,Y):= f(X,X,Y,Y),
        $let(gg: $int > $int,gg(Z):= ff(Z,Z),p(gg(i)) ) ) ).
```

... is equivalent to ...

```
tff(i_type,type,i: $int).
tff(f_type,type,f: ($int * $int * $int * $int) > $int).
tff(p_type,type,p: $int > $o ).
tff(let_tuple_3,axiom,p(f(i,i,i,i)) ).
```

Tuples can be used directly in let expressions. For example ...

```
tff(p_type,type,p: ($int * $int) > $o ).
tff(q_type,type,q: [$int,$int] > $o ).
tff(let_tuple_4,axiom,$let([a:$int,b:$int], [a,b]:= [27,28], p(a,b)) ).
tff(let_tuple_5,axiom,$let(d: [$int,$int], d:= [27,28],q(d)) ).
```

... is equivalent to ...

```
tff(p_type,type,p: ($int * $int) > $o ).
tff(q_type,type,q: [$int,$int] > $o ).
tff(let_tuple_4,axiom,p(27,28)) ).
tff(let_tuple_5,axiom,q([27,28])) ).
```

Tuples, conditional expressions, and let expressions can be mixed in useful ways. For example ...

```
tff(cc_type,type,cc: $int).
tff(dd_type,type,dd: $int).
tff(let_ite_1,axiom,
    $let([aa: $int,bb: $int],
        [aa,bb]:= $ite($greater(cc,dd),[cc,dd],[dd,cc]),
        p(aa,bb)) ).
```

4 Software Support and Examples

4.1 Software for TFX

BNF, TPTP4X

Vampire EVGENY

Voogie EVGENY

Mention BLT [3]

4.2 Examples

Figure 3 shows an example that uses formulae as terms, quantifies over boolean variables, and uses boolean equality as equivalence.

```
%-----
tff(impl,type, imply: ( $o * $o ) > $o ).
tff(graph,type, p: ( $i * $i ) > $o ).

tff(impl_definition,axiom,
    ! [X:$o,Y:$o] : ( imply(X,Y) = ( ~ X | Y ) ) ).

tff(graph_impl,axiom,
    ! [X: $i,Y: $i] : ( p(X,Y) => f(X) = Y ) ).

tff(graph_conjecture,conjecture,(
    ! [X: $i,Y: $i,Z: $i] : imply(p(X,Y) & p(X,Z),Y = Z ) ).
%-----
```

Figure 3: Implication is a function

Figure 4 shows the TFX encoding of the FOOL formula on Figure 2 that expresses a partial correctness property of an imperative program with an `if` statement.

```
%-----
tff(x,type,x:$int).
tff(y,type,y:$int).
tff(t,type,t:$int).
tff(x_leq_y,conjecture,
    $let([x:$int,y:$int,t:$int],
        [x,y,t] := $ite($greater(x,y),
            $let(t:$int, t := x,
                $let(x:$int, x := y,
                    $let(y:$int, y := t,
                        [x,y,t]))),
            [x,y,t]),
        $lesseq(x,y))).
%-----
```

Figure 4: A TFX encoding of a program analysis problem on Figure 2.

Figure 5 shows an example that uses formulae as terms, in the second arguments of the `says` predicate. The problem is to find a model, from the it is possible to determine which of A,B, or C is the only truth teller on this Smullyanesque island [9].

5 Conclusion

This paper has described

Acknowledgements. Thanks to

```

%-----
tff(a_type,type, a: $i ).
tff(b_type,type, b: $i ).
tff(c_type,type, c: $i ).
tff(exactly_one_truthteller_type,type, exactly_one_truthteller: $o ).
tff(says,type, says: ( $i * $o ) > $o ).

%---Each person is either a truthteller or a liar
tff(island,axiom,
  ! [P: $i] :
    ( says(P: $i,$true) <~> says(P: $i,$false) ) ).
tff(exactly_one_truthteller,axiom,
  ( exactly_one_truthteller
    <=> ( ? [P: $i] : says(P,$true)
      & ! [P1: $i,P2: $i] :
        ( ( says(P1,$true) & says(P2,$true) )
          => P1 = P2 ) ) ) ).

%---B said that A said that there is exactly one truthteller on the island
tff(b_says,hypothesis, says(b,says(a,exactly_one_truthteller)) ).

%---C said that what B said is false
tff(c_says,hypothesis, says(c,says(b,$false)) ).
%-----

```

Figure 5: Who is the truthteller?

References

- [1] C. Barrett, A. Stump, and C. Tinelli. The SMT-LIB Standard: Version 2.0. In A. Gupta and D. Kroening, editors, *Proceedings of the 8th International Workshop on Satisfiability Modulo Theories*, 2010.
- [2] J. Blanchette and A. Paskevich. TFF1: The TPTP Typed First-order Form with Rank-1 Polymorphism. In M.P. Bonacina, editor, *Proceedings of the 24th International Conference on Automated Deduction*, number 7898 in Lecture Notes in Artificial Intelligence, pages 414–420. Springer-Verlag, 2013.
- [3] YuTing Chen and Carlo A. Furia. Triggerless happy – intermediate verification with a first-order prover. In Nadia Polikarpova and Steve Schneider, editors, *Proceedings of the 13th International Conference on integrated Formal Methods (iFM)*, volume 10510 of *Lecture Notes in Computer Science*, pages 295–311. Springer, September 2017.
- [4] C. Kaliszyk, G. Sutcliffe, and F. Rabe. TH1: The TPTP Typed Higher-Order Form with Rank-1 Polymorphism. In P. Fontaine, S. Schulz, and J. Urban, editors, *Proceedings of the 5th Workshop on the Practical Aspects of Automated Reasoning*, number 1635 in CEUR Workshop Proceedings, pages 41–55, 2016.
- [5] E. Kotelnikov, L. Kovacs, G. Reger, and A. Voronkov. The Vampire and the FOOL. In J. Avigad and A. Chlipala, editors, *Proceedings of the 5th ACM SIGPLAN Conference on Certified Programs and Proofs*, pages 37–48. ACM, 2016.
- [6] E. Kotelnikov, L. Kovacs, M. Suda, and A. Voronkov. A Clausal Normal Form Translation for FOOL. In C. Benzmüller, G. Sutcliffe, and R. Rojas, editors, *Proceedings of the 2nd Global Conference on Artificial Intelligence*, number 41 in EPiC Series in Computing, pages 53–71, 2016.

- [7] E. Kotelnikov, L. Kovacs, and A. Voronkov. A First Class Boolean Sort in First-Order Theorem Proving and TPTP. In M. Kerber, J. Carette, C. Kaliszyk, F. Rabe, and V. Sorge, editors, *Proceedings of the International Conference on Intelligent Computer Mathematics*, number 9150 in Lecture Notes in Computer Science, pages 71–86. Springer-Verlag, 2015.
- [8] E. Kotelnikov, L. Kovacs, and A. Voronkov. A FOOLish Encoding of the Next State Relations of Imperative Programs. In D. Galmiche, S. Schulz, and R. Sebastiani, editors, *Proceedings of the 9th International Joint Conference on Automated Reasoning*, page Submitted, 2018.
- [9] R.M. Smullyan. *What is the Name of This Book? The Riddle of Dracula and Other Logical Puzzles*. Prentice-Hall, 1978.
- [10] G. Sutcliffe. The TPTP Problem Library and Associated Infrastructure. The FOF and CNF Parts, v3.5.0. *Journal of Automated Reasoning*, 43(4):337–362, 2009.
- [11] G. Sutcliffe. The TPTP World - Infrastructure for Automated Reasoning. In E. Clarke and A. Voronkov, editors, *Proceedings of the 16th International Conference on Logic for Programming, Artificial Intelligence, and Reasoning*, number 6355 in Lecture Notes in Artificial Intelligence, pages 1–12. Springer-Verlag, 2010.
- [12] G. Sutcliffe and C. Benzmüller. Automated Reasoning in Higher-Order Logic using the TPTP THF Infrastructure. *Journal of Formalized Reasoning*, 3(1):1–27, 2010.
- [13] G. Sutcliffe, S. Schulz, K. Claessen, and P. Baumgartner. The TPTP Typed First-order Form with Arithmetic. In N. Bjørner and A. Voronkov, editors, *Proceedings of the 18th International Conference on Logic for Programming, Artificial Intelligence, and Reasoning*, number 7180 in Lecture Notes in Artificial Intelligence, pages 406–419. Springer-Verlag, 2012.
- [14] G. Sutcliffe and C.B. Suttner. The TPTP Problem Library: CNF Release v1.2.1. *Journal of Automated Reasoning*, 21(2):177–203, 1998.